

**irf ograph** — Overlaid graphs of IRFs, dynamic-multiplier functions, and FEVDs

Description  
Options

Quick start  
Remarks and examples

Menu  
Stored results

Syntax  
Also see

## Description

`irf ograph` displays plots of `irf` results on one graph (one pair of axes).

## Quick start

Graph of an orthogonalized IRF `myirf` overlaid on cumulative IRF `mycirf` for dependent variables `y1` and `y2`

```
irf ograph (myirf y1 y2 oirf) (mycirf y1 y2 cirf)
```

Same as above, and include confidence bands and add a title

```
irf cgraph (myirf y1 y2 oirf) (mycirf y1 y2 cirf), ci ///
title("My Title")
```

Note: `irf` commands can be used after `var`, `svar`, `ivsvar`, `vec`, `arma`, `arfima`, `lpirf`, `dsge`, or `dsgenl`; see [TS] [var](#), [TS] [var svar](#), [TS] [var ivsvar](#), [TS] [vec](#), [TS] [arma](#), [TS] [arfima](#), [TS] [lpirf](#), [DSGE] [dsge](#), or [DSGE] [dsgenl](#).

## Menu

Statistics > Postestimation

## Syntax

```
irf ograph (spec1) [ (spec2) ... [ (spec15) ] ] [ , options ]
```

where (*spec*<sub>*k*</sub>) is

```
(irfname impulsevar responsevar stat [ , plot_options ])
```

*irfname* is the name of a set of IRF results in the active IRF file or “.”, which means the first named result in the active IRF file. *impulsevar* should be specified as an endogenous variable for all statistics except *dm* and *cdm*; for those, specify as an exogenous variable. *responsevar* is an endogenous variable name. *stat* is one or more statistics from the list below:

<i>stat</i>	Description
-------------	-------------

Main	
<i>irf</i>	impulse–response function
<i>oirf</i>	orthogonalized impulse–response function
<i>dm</i>	dynamic-multiplier function
<i>cirf</i>	cumulative impulse–response function
<i>coirf</i>	cumulative orthogonalized impulse–response function
<i>cdm</i>	cumulative dynamic-multiplier function
<i>fevd</i>	Cholesky forecast-error variance decomposition
<i>sirf</i>	structural impulse–response function
<i>sfevd</i>	structural forecast-error variance decomposition

<i>plot_options</i>	Description
---------------------	-------------

Main	
<i>ci</i>	add confidence bands to the graph
<i>level</i> (#)	set confidence level; default is <i>level</i> (95)
<i>lstep</i> (#)	use # for first step
<i>ustep</i> (#)	use # for maximum step

Plot	
<i>cline_options</i>	affect rendition of the plotted lines

CI plot	
<i>ciopts</i> ( <i>area_options</i> )	affect rendition of the confidence intervals

*plot\_options* may be specified within a plot specification, globally, or in both. When specified in a plot specification, *plot\_options* affect only the specification in which they are used. When supplied globally, *plot\_options* affect all plot specifications. When supplied in both places, options in the plot specification take precedence.

<i>options</i>	Description
----------------	-------------

<i>set</i> ( <i>filename</i> )	make <i>filename</i> active
--------------------------------	-----------------------------

Graph options	
<i>graph_options</i>	any options other than <i>by</i> () documented in [G-3] <i>twoway_options</i>

*collect* is allowed; see [U] 11.1.10 **Prefix commands**.

## Options

`set(filename)` specifies the file to be made active; see [TS] [irf set](#). If `set()` is not specified, the active file is used.

*plot\_options* affect the rendition of the specification and include the following:

### Main

`ci` adds confidence bands to the graph. The `noc` option may be used within a plot specification to suppress its confidence bands when the `ci` option is supplied globally.

`level(#)` specifies the confidence level, as a percentage, for confidence bands; see [U] [20.8 Specifying the width of confidence intervals](#).

`lstep(#)` specifies the first step, or period, to be included in the graph. `lstep(0)` is the default.

`ustep(#)`,  $\# \geq 1$ , specifies the maximum step, or period, to be included.

### Plot

*cline\_options* affect the rendition of the plotted lines; see [G-3] [cline\\_options](#).

### CI plot

`ciopts(area_options)` affects the rendition of the confidence bands for the plotted statistic; see [G-3] [area\\_options](#). `ciopts()` implies `ci`.

### Graph options

*graph\_options* are any of the options documented in [G-3] [twoway\\_options](#), excluding `by()`. These include options for titling the graph (see [G-3] [title\\_options](#)) and for saving the graph to disk (see [G-3] [saving\\_option](#)).

## Remarks and examples

[stata.com](http://www.stata.com)

If you have not read [TS] [irf](#), please do so.

`irf ograph` overlays plots of IRFs and FEVDs on one graph.


### ► Example 1

We have previously issued the commands:

```
. use https://www.stata-press.com/data/r18/lutkepohl2
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4), lags(1/2) dfk
. irf create order1, step(10) set(myirf1, new)
. irf create order2, step(10) order(dln_inc dln_inv dln_consump)
```

We now wish to compare the `oirf` for impulse `dln_inc` and response `dln_consump` for two different Cholesky orderings:

```
. irf ograph (order1 dln_inc dln_consump oirf)
> (order2 dln_inc dln_consump oirf)
```




◀

#### □ Technical note

Graph options allow you to change the appearance of each plot. The following graph contains the plots of the FEVDs for impulse `dln_inc` and each response using the results from the first collection of results in the active IRF file (using the “.” shortcut). We specify `ciopts(color())` with each plot to request confidence bands and specify a color that corresponds to the plotted line. In the second plot, we supply the `clpat(dash)` option (an abbreviation for `clpattern(dash)`) to give the line a dashed pattern. In the third plot, we supply the `m(o)`, `clpat(dash_dot)`, and `recast(connected)` options to get small circles connected by a line with a dash-dot pattern. We use the `title()` option to add a descriptive title to the graph.

```
. irf ograph (. dln_inc dln_inc fevd, ciopts(color(ltblue)))
> (. dln_inc dln_consump fevd, ciopts(color(cranberry*.3))
>      clpat(dash))
> (. dln_inc dln_inv fevd, ciopts(color(green*.3))
>      m(o) clpat(dash_dot) recast(connected))
> , title("Comparison of forecast-error variance decomposition")
```



The `clpattern()` option is described in [G-3] [connect\\_options](#), `msymbol()` is described in [G-3] [marker\\_options](#), `title()` is described in [G-3] [title\\_options](#), and `recast()` is described in [G-3] [advanced\\_options](#). □

## Stored results

`irf ograph` stores the following in `r()`:

Scalars

`r(plots)`            number of plot specifications  
`r(ciplots)`          number of plotted confidence bands

Macros

`r(irfname#)`        *irfname* from (*spec#*)  
`r(impulse#)`        impulse from (*spec#*)  
`r(response#)`       response from (*spec#*)  
`r(stat#)`            statistics from (*spec#*)  
`r(ci#)`              level from (*spec#*) or *nocl*

## Also see

[TS] [irf](#) — Create and analyze IRFs, dynamic-multiplier functions, and FEVDs

[TS] [var intro](#) — Introduction to vector autoregressive models

[TS] [vec intro](#) — Introduction to vector error-correction models

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).